
À propos

You can share this PDF with anyone you feel could benefit from it, downloaded the latest version 
from: ada

It is an unofficial and free ada ebook created for educational purposes. All the content is extracted 
from Stack Overflow Documentation, which is written by many hardworking individuals at Stack 
Overflow. It is neither affiliated with Stack Overflow nor official ada.

The content is released under Creative Commons BY-SA, and the list of contributors to each 
chapter are provided in the credits section at the end of this book. Images may be copyright of 
their respective owners unless otherwise specified. All trademarks and registered trademarks are 
the property of their respective company owners.

Use the content presented in this book at your own risk; it is not guaranteed to be correct nor 
accurate, please send your feedback and corrections to info@zzzprojects.com

Chapitre 1: Démarrer avec ada

Remarques

Ada est un langage de programmation informatique de haut niveau, orienté objet et standardisé 
au niveau international, qui prend en charge les typages forts et la programmation structurée. Plus 
d'informations peuvent être trouvées ici .

Versions

Version Date de sortie

Ada 2012 (TC-1) 2016-04-01

Ada 2012 2012-12-10

Ada 2005 2007-01-01

Ada 95 1995-12-10

Ada 83 1983-01-01

Examples

Installation ou configuration

Ada est un langage de programmation pour lequel il existe plusieurs compilateurs.

L'un de ces compilateurs, et peut-être le plus utilisé, est GNAT. Cela fait partie de la chaîne 
d'outils GCC. Il peut être installé à partir de plusieurs sources:

La version annuelle de la GPL réalisée par AdaCore, disponible gratuitement sur le site 
libre . Cette version a subi tous les tests internes effectués par AdaCore pour ses 
versions professionnelles, est disponible sur un grand nombre de plates-formes. Le 
compilateur et son exécution sont publiés sous licence GPL et, à moins que vous 
n'utilisiez aucun runtime, tous les exécutables que vous distribuez seront également 
couverts par cette licence. Pour les universitaires et les projets à leurs débuts, ce n’est 
pas un problème.

○

La FSF gcc reçoit régulièrement les mêmes correctifs. La version de GNAT n'est peut-
être pas toujours à jour, mais se rattrape régulièrement.

○

Un certain nombre de contributeurs regroupent cette version de FSF pour diverses 
distributions Linux (systèmes basés sur Debian, entre autres) et des binaires pour Mac 
OS X. L'utilisation du gestionnaire de paquets depuis votre distribution pourrait être le 

○

•

moyen le plus simple d'installer GNAT. Ces versions sont fournies avec la licence GCC 
standard et vous permettent d'écrire du code source fermé.

AdaCore fournit également GNAT Pro , qui est fourni avec la licence GCC standard qui 
vous permet d’écrire du code source fermé. Plus important peut-être, il est fourni avec 
un support, si vous avez des questions sur l'utilisation du langage, des outils, comment 
mieux implémenter quelque chose, et bien sûr des rapports de bogues et des 
demandes d'amélioration.

○

Un autre nombre de compilateurs sont répertoriés dans le WikiBook d'Ada , ainsi que des 
instructions d'installation. Getadanow.com propose des éditions de FSF GNAT, prêtes à l'emploi 
pour différents systèmes d'exploitation sur plusieurs types de matériel ou de machines virtuelles. 
Le site recueille également des ressources pour apprendre et partager Ada.

Bonjour le monde

with Ada.Text_IO; 
 
procedure Hello_World is 
begin 
   Ada.Text_IO.Put_Line ("Hello World"); 
end Hello_World;

Alternativement, après avoir importé le paquet Ada.Text_IO , vous pouvez dire use Ada.Text_IO; 
afin de pouvoir utiliser Put_Line sans déclarer explicitement de quel paquet il doit provenir, en tant 
que tel:

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Hello_World is 
begin 
    Put_Line ("Hello World"); 
end Hello_World;

Si vous utilisez le compilateur gnat , ce programme simple peut être compilé avec

gnatmake hello_world

Cela générera un certain nombre de fichiers, y compris un hello_world (ou hello_world.exe sous 
Windows) que vous pourrez exécuter pour voir le message célèbre. Le nom de l'exécutable est 
calculé automatiquement à partir du nom du sous-programme principal Ada. Dans Ada, un sous-
programme principal peut avoir un nom quelconque. Il suffit que ce soit une procédure sans 
paramètre que vous donnez comme argument à gnatmake .

D'autres compilateurs ont des exigences similaires, bien que la commande de construction soit 
bien sûr différente.

Version

Le langage de programmation Ada standard est défini dans le manuel de référence Ada . Les 
changements de version provisoires et les notes de version sont discutés dans les documents de 
justification correspondants. Les implémentations documentent généralement leur conformité à la 
norme sous la forme d'un guide d'utilisation et / ou d'un manuel de référence, par exemple .

Ada 2012

Manuel de référence du langage Ada 2012○

Justification de Ada 2012○

•

Ada 2005

Ada 2005 Language Reference Manual○

Raison d'être d'Ada 2005○

•

Ada 95

Ada 95 Language Reference Manu l○

Justification pour Ada 95○

•

Ada 83

Ada 83 Référence du langage Manua l○

Ada 83 Raison d'être de la conception du langage de programmation Ada®○

•

Bibliothèques

Comme pour tout langage de programmation, Ada est livré avec de nombreuses bibliothèques 
pour accomplir diverses tâches. Voici quelques conseils pour certains d’entre eux, bien que la 
recherche sur github en mènera plus.

Le runtime Ada lui-même, distribué à tous les compilateurs, comprend un ensemble complet 
de packages et d’annexes, allant des structures de données et des conteneurs aux entrées / 
sorties, à la manipulation de chaînes, aux manipulations de temps, aux commutateurs de 
ligne de commande, nombres aléatoires,...

•

Le compilateur GNAT est livré avec son propre runtime étendu, avec de nouveaux paquets 
dans la hiérarchie GNAT , qui prennent en charge les expressions régulières, le tri, la 
recherche, l'unicode, le CRC, l'heure d'entrée / sortie, ...

•

gnatcoll est une bibliothèque disponible sur le site libre d'AdaCore. Elle comprend une 
infrastructure de journalisation étendue, des applications étendues avec python, mmap, une 
infrastructure complète d'interface avec les systèmes de fichiers, l'analyse des e-mails et 
des boîtes aux lettres. de manière sécurisée, interface avec diverses bibliothèques telles 
que l'icône, readline, couleurs du terminal, prise en charge des types comptés de référence 
pour la gestion automatique de la mémoire, fichiers JSON, ...

•

XML / Ada est une bibliothèque pour analyser et valider des documents XML•

GtkAda est une liaison complète à la bibliothèque gtk +, qui vous permet d'écrire des •

interfaces utilisateur portables sur Unix, Windows et OSX.

AWS est une structure permettant de créer des serveurs Web dans Ada, avec une prise en 
charge complète de divers protocoles tels que HTTP, Websockets, ... et son propre système 
de modèles.

•

Lire Démarrer avec ada en ligne: https://riptutorial.com/fr/ada/topic/3900/demarrer-avec-ada

Chapitre 2: Énumération

Syntaxe

function Enumeration ' Image (Argument: Enumeration'Base) retourne String;•
function Enumeration ' Img return String; - GNAT•
function Enumeration'Val (Argument: Universal_Integer) retourne Enumeration'Base;•
function Enumeration'Pos (Argument: Enumeration'Base) retourne Universal_Integer;•
function Enumeration'Enum_Rep (Argument: Enumeration'Base) retourne Universal_Integer;•
function Literal 'Enum_Rep retourne Universal_Integer; - GNAT•
function Literal 'Retour d'adresse System.Address;•
pour Enumeration use (Literal_1 => Universal_Integer , Literal_n => Universal_Integer );•
( Littéral en Enumération) renvoyer Booléen;•

Examples

Littérature itérative

Un littéral à l'intérieur d'une énumération est un type discret, nous pouvons donc utiliser l'attribut 
Image pour déterminer le littéral sous forme de texte. Notez que cela imprime le même mot que 
dans le code (mais en majuscule).

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   type Fruit is (Banana, Pear, Orange, Melon); 
begin 
   for I in Fruit loop 
      Put (Fruit'Image (I)); 
      New_Line; 
   end loop; 
end;

Résultat

BANANA 
PEAR 
ORANGE 
MELON

Utilisation du package Enumeration_IO

Au lieu d'attribuer Image et Ada.Text_IO.Put aux littéraux d'énumération, nous ne pouvons utiliser 
que le package générique Ada.Text_IO.Enumeration_IO pour imprimer les littéraux.

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   type Fruit is (Banana, Pear, Orange, Melon); 
   package Fruit_IO is new Enumeration_IO (Fruit); use Fruit_IO; 
begin 
   for I in Fruit loop 
      Put (I); 
      New_Line; 
   end loop; 
end;

Résultat

BANANA 
PEAR 
ORANGE 
MELON

Caractères majuscules

L'attribut Image majuscule tous les caractères des littéraux d'énumération. La fonction 
Case_Rule_For_Names applique des majuscules pour le premier caractère et rend le reste en 
minuscule.

with Ada.Text_IO; use Ada.Text_IO; 
with Ada.Strings.Maps.Constants; use Ada.Strings.Maps.Constants; 
with Ada.Strings.Fixed; use Ada.Strings.Fixed; 
 
procedure Main is 
   type Fruit is (Banana, Pear, Orange, Melon); 
   function Case_Rule_For_Names (Item : String) return String is 
   begin 
      return Translate (Item (Item'First .. Item'First), Upper_Case_Map) & Translate (Item 
(Item'First + 1 .. Item'Last), Lower_Case_Map); 
   end; 
begin 
   for I in Fruit loop 
      Put (Case_Rule_For_Names (Fruit'Image (I))); 
      New_Line; 
   end loop; 
end;

Résultat

Banana 
Pear 
Orange 
Melon

Cas de titre, Utilisation de Enumeration_IO, Pour une sous-gamme

Combinaison du changement de casse avec Enumeration_IO et utilisation d'un tampon de texte 
pour l'image. Le premier caractère est manipulé en place.

with Ada.Text_IO; use Ada.Text_IO; 
with Ada.Characters.Handling; use Ada.Characters.Handling; 
 
procedure Main is 
    type Fruit is (Banana, Pear, Orange, Melon); 
    package Fruit_IO is new Enumeration_IO (Fruit); 
    Buffer : String (1 .. Fruit'Width); 
begin 
   for I in Fruit range Pear .. Fruit'Last loop 
       Fruit_IO.Put (To => Buffer, 
                     Item => I, 
                     Set => Lower_Case); 
       Buffer (Buffer'First) := To_Upper (Buffer (Buffer'First)); 
       Put_Line (Buffer); 
   end loop; 
end;

Résultat

Pear 
Orange 
Melon 

Lire Énumération en ligne: https://riptutorial.com/fr/ada/topic/5930/enumeration

Chapitre 3: Fichiers et flux d'E / S

Remarques

La bibliothèque standard Ada fournit des E / S de fichiers traditionnels de données textuelles ou 
binaires, ainsi que des E / S de fichiers en streaming. Les fichiers de données binaires seront des 
séquences de valeurs d'un type, tandis que les fichiers de flux peuvent être des séquences de 
valeurs de types éventuellement différents.

Pour lire et écrire des éléments de différents types depuis / vers des fichiers de flux, Ada utilise 
des sous-programmes désignés par des attributs de type, à savoir 'Read , 'Write , 'Input et 'Output 
. Les deux derniers liront et écrivent les limites des tableaux, les discriminants d’enregistrement et 
les balises de type, en plus de l’entrée et de la sortie que Read et 'Write .

Examples

Créer et écrire dans un fichier

Les procédures Create , Put_Line , Close du package Ada.Text_IO de créer et d'écrire dans le fichier 
file.txt .

with Ada.Text_IO; 
 
procedure Main is 
   use Ada.Text_IO; 
   F : File_Type; 
begin 
   Create (F, Out_File, "file.txt"); 
   Put_Line (F, "This string will be written to the file file.txt"); 
   Close (F); 
end;

Fichier résultant file.txt

This string will be written to the file.txt

Créer et écrire dans un flux

Les attributs orientés flux des sous-types sont appelés pour écrire des objets dans un fichier, en 
utilisant des représentations binaires par défaut.

with Ada.Streams.Stream_IO; 
 
procedure Main is 
   type Fruit is (Banana, Orange, Pear); 
   type Color_Value is range 0 .. 255; 

   type Color is record 
      R, G, B : Color_Value; 
   end record; 
 
   Fruit_Colors : constant array (Fruit) of Color := 
     (Banana => Color'(R => 243, G => 227, B => 18), 
      Orange => Color'(R => 251, G => 130, B => 51), 
      Pear   => Color'(R => 158, G => 181, B => 94)); 
 
   use Ada.Streams.Stream_IO; 
 
   F : File_Type; 
 
begin 
   Create (F, Name => "file.bin"); 
   for C in Fruit_Colors'Range loop 
      Fruit'Write (Stream (F), C); 
      Color'Write (Stream (F), Fruit_Colors (C)); 
   end loop; 
   Close (F); 
end Main;

Fichier résultant

00000000  00 2e f3 00 e3 00 12 00  01 2e fb 00 82 00 33 00 
00000010  02 2e 9e 00 b5 00 5e 00 

Ouvrir et lire à partir d'un fichier de flux

Lisez les données de Create And Write To A Stream dans un programme.

with Ada.Streams.Stream_IO; 
 
procedure Main is 
   -- 
   --  ... same type definitions as in referenced example 
   -- 
   Fruit_Colors : array (Fruit) of Color; 
 
   use Ada.Streams.Stream_IO; 
 
   F : File_Type; 
   X : Fruit; 
begin 
   Open (F, Mode => In_File, Name => "file.bin"); 
   loop 
      Fruit'Read (Stream (F), X); 
      Color'Read (Stream (F), Fruit_Colors (X)); 
   end loop; 
exception 
   when End_Error => 
      Close (F); 
   pragma Assert  -- check data are the same 
     (Fruit_Colors (Banana) = Color'(R => 243, G => 227, B => 18) and 
      Fruit_Colors (Orange) = Color'(R => 251, G => 130, B => 51) and 
      Fruit_Colors (Pear)   = Color'(R => 158, G => 181, B => 94)); 

end Main;

Lire Fichiers et flux d'E / S en ligne: https://riptutorial.com/fr/ada/topic/8865/fichiers-et-flux-d-e---s

Chapitre 4: Généralité à Ada

Examples

Sous-programmes génériques

Les sous-programmes génériques sont utiles pour créer des sous-programmes ayant la même 
structure pour plusieurs types. Par exemple, pour échanger deux objets:

generic 
    type A_Type is private; 
procedure Swap (Left, Right : in out A_Type) is 
    Temp : A_Type := Left; 
begin 
    Left := Right; 
    Right := Temp; 
end Swap;

Paquets génériques

Dans le package générique Ada, lors de l'instanciation, les données sont dupliquées; en d'autres 
termes, si elles contiennent des variables globales, chaque instance aura sa propre copie de la 
variable, correctement typée et indépendante des autres.

generic 
    type T is private; 
package Gen is 
    type C is tagged record 
        V : T; 
    end record; 
    G : Integer; 
end Gen;

Paramètres génériques

Ada offre une grande variété de paramètres génériques difficiles à traduire dans d'autres langues. 
Les paramètres utilisés lors de l'instanciation et par conséquent ceux sur lesquels l'unité 
générique peut s'appuyer peuvent être des variables, des types, des sous-programmes ou des 
instances de package, avec certaines propriétés. Par exemple, ce qui suit fournit un algorithme de 
tri pour tout type de tableau:

generic 
    type Component is private; 
    type Index is (<>); 
    with function "<" (Left, Right : Component) return Boolean; 
    type Array_Type is array (Index range <>) of Component; 
procedure Sort (A : in out Array_Type);

Lire Généralité à Ada en ligne: https://riptutorial.com/fr/ada/topic/9322/generalite-a-ada

Chapitre 5: Image d'attribut

Introduction

Les attributs de sous-type 'Image et 'Value prendront respectivement une valeur scalaire et une 
chaîne et renverront respectivement une chaîne et une valeur scalaire. Le résultat de 'Image peut 
être entré dans 'Value pour obtenir la valeur d'origine. L'inverse est également vrai.

L'attribut __Scalar_Object__'Image peut être utilisé directement sur les objets (depuis Ada 2012-TC-
1).

Syntaxe

function Scalar'Image (Argument: Scalar'Base) retourne String;•
function Discrete'Image (Argument: Discrete'Base) retourne String;•
function Integer'Image (Argument: Integer'Base) retourne String;•
function Enumeration'Image (Argument: Enumeration'Base) return String;•
function Real'Image (Argument: Real'Base) retourne String;•
function Numeric'Image (Argument: Numeric'Base) retourne String;•
fonction Scalar'Value (Argument: String) renvoie Scalar'Base;•
function Discrete'Value (Argument: String) retourne Discrete'Base;•
function Integer'Value (Argument: String) retourne Integer'Base;•
function Enumeration'Value (Argument: String) renvoie Enumeration'Base;•
function Real'Value (Argument: String) renvoie Real'Base;•
function Scalar_Object 'Image return String;•

Remarques

Notez que 'Image peut entraîner des résultats définis par l'implémentation (RM 3.5), à savoir 
lorsque certains caractères graphiques nécessaires au résultat String ne sont pas définis dans 
Character . Considérons les plus grands répertoires de 'Wide_Image et 'Wide_Wide_Image .

Ada 2012 (TC-1)

L'autorisation d'utiliser l'attribut __Scalar_Object__'Image directement sur un objet a été ajoutée 
dans Ada 2012-TC-1 (avril 2016).

Examples

Imprimez float en utilisant l'attribut Image

Ada 2012 (TC-1)

with Ada.Text_IO; 

procedure Main is 
   type Some_Float digits 8 range 0.0 .. 10.0; 
   X : Some_Float := 2.71; 
begin 
   Ada.Text_IO.Put_Line (X'Image); 
end Main;

Résultat

2.71000E+00

Imprimer un entier en utilisant l'attribut Image

Ada 2012 (TC-1)

with Ada.Text_IO; 
 
procedure Main is 
   type Some_Integer is range -42 .. 42; 
   X : Some_Integer := 17; 
begin 
   Ada.Text_IO.Put_Line (X'Image); 
end Main;

Résultat

17

Imprimer l'énumération à l'aide de l'attribut Image

Ada 2012 (TC-1)

with Ada.Text_IO; 
 
procedure Main is 
   type Fruit is (Banana, Orange, Pear); 
   X : Fruit := Orange; 
begin 
   Ada.Text_IO.Put_Line (X'Image); 
   Ada.Text_IO.Put_Line (Pear'Image); 
end Main;

Résultat

ORANGE 
PEAR

Imprimer l'énumération à l'aide de l'attribut Image

with Ada.Text_IO; 
 
procedure Main is 
   type Fruit is (Banana, Orange, Pear); 
   X : Fruit := Orange; 
begin 
   Ada.Text_IO.Put_Line (Fruit'Image (X)); 
end Main;

Résultat

ORANGE

Imprimer Integer en utilisant l'attribut Image

with Ada.Text_IO; 
 
procedure Main is 
   X : Integer := 17; 
begin 
   Ada.Text_IO.Put_Line (Integer'Image (X)); 
end Main;

Résultat

17

Imprimer Float en utilisant l'attribut Image

with Ada.Text_IO; 
 
procedure Main is 
   X : Float := 2.71; 
begin 
   Ada.Text_IO.Put_Line (Float'Image (X)); 
end Main;

Résultat

2.71000E+00

En tant qu'inverses

with Ada.Text_IO; 
 
procedure Image_And_Value is 
   type Fruit is (Banana, Orange, Pear); 
   X  : Fruit := Orange; 
begin 
   Ada.Text_IO.Put_Line (Boolean'Image 
      (Fruit'Value (Fruit'Image (X)) = X 
          and 
       Fruit'Image (Fruit'Value ("ORANGE")) = "ORANGE")); 
end Image_And_Value;

Résultat

TRUE

Lire Image d'attribut en ligne: https://riptutorial.com/fr/ada/topic/4290/image-d-attribut

Chapitre 6: Mise en œuvre du modèle 
producteur-consommateur

Introduction

Une démonstration de la manière dont le modèle producteur-consommateur est mis en œuvre 
dans Ada.

Syntaxe

function Scalar ' Image (Argument: Scalar'Base) retourne String;•
tâche nom_tâche;•
task TaskName est la fin des entrées;•
le corps de la tâche Task_Name est les déclarations commencent fin du code;•
entry Nom_entrée;•
accepter Nom_Entrée;•
Sortie;•

Remarques

Les exemples doivent tous garantir une terminaison de tâche correcte.

Examples

Utiliser un tampon synchronisé

with Ada.Containers.Synchronized_Queue_Interfaces; 
with Ada.Containers.Unbounded_Synchronized_Queues; 
with Ada.Text_IO; 
 
procedure Producer_Consumer_V1 is 
   type Work_Item is range 1 .. 100; 
 
   package Work_Item_Queue_Interfaces is 
     new Ada.Containers.Synchronized_Queue_Interfaces 
           (Element_Type => Work_Item); 
 
   package Work_Item_Queues is 
     new Ada.Containers.Unbounded_Synchronized_Queues 
           (Queue_Interfaces => Work_Item_Queue_Interfaces); 
 
   Queue : Work_Item_Queues.Queue; 
 
   task type Producer; 
   task type Consumer; 
 
   Producers : array (1 .. 1)  of Producer; 

   Consumers : array (1 .. 10) of Consumer; 
 
   task body Producer is 
   begin 
      for Item in Work_Item loop 
         Queue.Enqueue (New_Item => Item); 
      end loop; 
   end Producer; 
 
   task body Consumer is 
      Item : Work_Item; 
   begin 
      loop 
         Queue.Dequeue (Element => Item); 
         Ada.Text_IO.Put_Line (Work_Item'Image (Item)); 
      end loop; 
   end Consumer; 
 
begin 
   null; 
end Producer_Consumer_V1;

Notez que je suis paresseux ici: il n'y a pas de fin correcte des tâches du consommateur, une fois 
que tous les éléments de travail sont consommés.

Modèle Producteur-Consommateur utilisant le mécanisme Ada Rendezvous

Une solution synchrone producteur-consommateur garantit que le consommateur lit chaque 
donnée écrite par le producteur exactement une fois. Les solutions asynchrones permettent au 
consommateur d’échantillonner la sortie du producteur. Soit le consommateur consomme les 
données plus rapidement qu'il n'est produit, soit le consommateur consomme les données plus 
lentement qu'il n'est produit. L'échantillonnage permet au consommateur de gérer les données 
actuellement disponibles. Ces données peuvent n'être qu'un échantillonnage des données 
produites ou des données déjà consommées.

------------------------------------------------------------------ 
-- synchronous PC using Rendezvous -- 
------------------------------------------------------------------ 
with Ada.Text_IO; use Ada.Text_IO; 
 
procedure PC_Rendezvous is 
   task Producer; 
   task Consumer is 
      entry Buf(Item : in Integer); 
   end Consumer; 
   task body Producer is 
   begin 
      for I in 1..10 loop 
         Put_Line("Producer writing" & Integer'Image(I)); 
         Consumer.Buf(I); 
      end loop; 
   end Producer; 
   task body Consumer is 
      Temp : Integer; 
   begin 
      loop 

         select 
            accept Buf(Item : in Integer) do 
               temp := Item; 
            end; 
            Put_Line("Consumer read" & Integer'Image(Temp)); 
         or 
            terminate; 
         end select; 
      end loop; 
   end Consumer; 
 
begin 
   null; 
end PC_Rendezvous;

Producteur-Consommateur avec un consommateur échantillonneur

Cet exemple utilise la procédure principale comme tâche du producteur. Dans Ada, la procédure 
principale s'exécute toujours dans une tâche distincte de toutes les autres tâches du programme, 
voir exemple minimal .

------------------------------------------------------------------ 
-- Sampling Consumer -- 
------------------------------------------------------------------ 
with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Sampling_PC is 
   protected Buf is 
      procedure Write(Item : in Integer); 
      function Read return Integer; 
      procedure Set_Done; 
      function Get_Done return Boolean; 
   private 
      Value : Integer := Integer'First; 
      Is_Done : Boolean := False; 
   end Buf; 
   protected body Buf is 
      procedure Write(Item : in Integer) is 
      begin 
         Value := Item; 
      end Write; 
      function Read return Integer is 
      begin 
         return Value; 
      end Read; 
      procedure Set_Done is 
      begin 
         Is_Done := True; 
      end Set_Done; 
      function Get_Done return Boolean is 
      begin 
         return Is_Done; 
      end Get_Done; 
   end Buf; 
 
   task Consumer; 
   task body Consumer is 
   begin 

      while not Buf.Get_Done loop 
          Put_Line("Consumer read" & Integer'Image(Buf.Read)); 
      end loop; 
   end Consumer; 
 
begin 
   for I in 1..10 loop 
     Put_Line("Producer writing" & Integer'Image(I)); 
     Buf.Write(I); 
   end loop; 
   Buf.Set_Done; 
end Sampling_PC;

Plusieurs producteurs et consommateurs partageant le même tampon

Cet exemple montre plusieurs producteurs et consommateurs partageant le même tampon. Les 
entrées protégées dans Ada implémentent une file d'attente pour gérer les tâches en attente. La 
politique de mise en file d'attente par défaut est First In First Out.

------------------------------------------------------------------ 
-- Multiple producers and consumers sharing the same buffer -- 
------------------------------------------------------------------ 
with Ada.Text_IO; use Ada.Text_Io; 
 
procedure N_Prod_Con is 
   protected Buffer is 
      Entry Write(Item : in Integer); 
      Entry Read(Item : Out Integer); 
   private 
      Value  : Integer := Integer'Last; 
      Is_New : Boolean := False; 
   end Buffer; 
   protected body Buffer is 
      Entry Write(Item : in Integer) when not Is_New is 
      begin 
         Value := Item; 
         Is_New := True; 
      end Write; 
      Entry Read(Item : out Integer) when Is_New is 
      begin 
         Item := Value; 
         Is_New := False; 
      end Read; 
   end Buffer; 
 
   task type Producers(Id : Positive) is 
      Entry Stop; 
   end Producers; 
   task body Producers is 
      Num : Positive := 1; 
   begin 
      loop 
         select 
            accept Stop; 
            exit; 
         or 
            delay 0.0001; 
         end select; 

         Put_Line("Producer" & Integer'Image(Id) & " writing" & Integer'Image(Num)); 
         Buffer.Write(Num); 
         Num := Num + 1; 
      end loop; 
   end Producers; 
 
   task type Consumers(Id : Positive) is 
      Entry Stop; 
   end Consumers; 
 
   task body Consumers is 
      Num : Integer; 
   begin 
      loop 
         select 
            accept stop; 
            exit; 
         or 
            delay 0.0001; 
         end select; 
         Buffer.Read(Num); 
         Put_Line("Consumer" & Integer'Image(ID) & " read" & Integer'Image(Num)); 
      end loop; 
   end Consumers; 
   P1 : Producers(1); 
   P2 : Producers(2); 
   P3 : Producers(3); 
   C1 : Consumers(1); 
   C2 : Consumers(2); 
   C3 : Consumers(3); 
begin 
   delay 0.2; 
   P1.Stop; 
   P2.Stop; 
   P3.Stop; 
   C1.Stop; 
   C2.Stop; 
   C3.Stop; 
end N_Prod_Con;

Lire Mise en œuvre du modèle producteur-consommateur en ligne: 
https://riptutorial.com/fr/ada/topic/8632/mise-en-ouvre-du-modele-producteur-consommateur

Chapitre 7: package Ada.Text_IO

Introduction

Package Ada.Text_IO est utilisé pour mettre du texte ou obtenir du texte à partir de fichiers ou de 
consoles.

Examples

Put_Line

Imprime une chaîne avec une nouvelle ligne.

with Ada.Text_IO; 
 
procedure Put_Text is 
    use Ada.Text_IO; 
    S : String := "Hello"; 
begin 
    Put_Line ("Hello"); 
    Put_Line (Standard_Output, "Hello"); 
    Put_Line (Standard_Error, "Hello error"); 
    Put_Line (S & " World"); 
end;

Résultat

Hello 
Hello 
Hello error 
Hello World

Lire package Ada.Text_IO en ligne: https://riptutorial.com/fr/ada/topic/8839/package-ada-text-io

Chapitre 8: Paquets

Syntaxe

avec Package_Name_To_Include;•
package New_Package_Name renomme Package_To_Rename;•
utilise le nom du paquet;•
package Parent_Name.Child_Name est•

Remarques

Forfait fournit:

Encapsulation de code•
Compilation séparée•
Masquer les procédures, les fonctions, les opérateurs sur les types privés•

Similitudes ou analogues dans d'autres langues:

Espace de noms C ++•
Packages Java•

Examples

Plus sur les forfaits

Dans Hello World , vous avez Ada.Text_IO le package Ada.Text_IO et comment l'utiliser pour 
effectuer des opérations d'E / S dans votre programme. Les paquets peuvent être manipulés pour 
faire beaucoup de choses différentes.

Renommer : pour renommer un package, vous utilisez le mot-clé renames dans une déclaration de 
package, en tant que tel:

package IO renames Ada.Text_IO;

Maintenant, avec le nouveau nom, vous pouvez utiliser la même notation en pointillés pour des 
fonctions telles que Put_Line (par exemple, IO.Put_Line ), ou vous pouvez simplement l' use avec 
use IO . Bien sûr, dire use IO ou IO.Put_Line utilisera les fonctions du package Ada.Text_IO .

Visibilité et isolation : Dans l'exemple Hello World, nous avons inclus le package Ada.Text_IO à 
l'aide d'une clause with . Mais nous avons également déclaré que nous voulions use Ada.Text_IO 
sur la même ligne. L' use Ada.Text_IO déclaration use Ada.Text_IO aurait pu être déplacée dans la 
partie déclarative de la procédure:

with Ada.Text_IO; 
 
procedure hello_world is 
   use Ada.Text_IO; 
begin 
   Put_Line ("Hello, world!"); 
end hello_world;

Dans cette version, les procédures, fonctions et types d' Ada.Text_IO sont directement disponibles 
dans la procédure. En dehors du bloc dans lequel Ada.Text_IO est déclaré, il faudrait utiliser la 
notation en pointillés pour invoquer, par exemple:

with Ada.Text_IO; 
 
procedure hello_world is 
begin 
   Ada.Text_IO.Put ("Hello, ");       --  The Put function is not directly visible here 
   declare 
      use Ada.Text_IO; 
   begin 
      Put_Line ("world!");            --  But here Put_Line is, so no Ada.Text_IO. is needed 
   end; 
end hello_world;

Cela nous permet d'isoler l'utilisation… les déclarations à l'endroit où elles sont nécessaires.

Relation parent-enfant

Afin de subdiviser les programmes Ada, les packages peuvent comporter des enfants. Celles-ci 
peuvent être des paquets aussi. Un package enfant a un privilège spécial: il peut voir les 
déclarations dans la partie privée du package parent. Une utilisation typique de cette visibilité 
spéciale consiste à former une hiérarchie de types dérivés dans la programmation orientée objet.

package Orders is 
   type Fruit is (Banana, Orange, Pear); 
   type Money is delta 0.01 digits 6; 
 
   type Bill is tagged private; 
 
   procedure Add 
     (Slip   : in out Bill; 
      Kind   : in     Fruit; 
      Amount : in     Natural); 
 
   function How_Much (Slip : Bill) return Money; 
 
   procedure Pay 
     (Ordered : in out Bill; 
      Giving  : in     Money); 
 
private 
   type Bill is tagged record 
      --  ... 
      Sum : Money := 0.0; 
   end record; 
end Orders;

Toute unité Ada dirigée par with Orders; peut déclarer des objets de type Bill , puis appeler les 
opérations Add , How_Much et Pay . Cependant, il ne voit pas les composants de Bill , ni même de 
Orders.Bill , car la définition de type complète est masquée dans la partie privée de Orders . La 
définition complète n'est cependant pas masquée dans les packages enfants. Cette visibilité 
facilite l'extension de type si nécessaire. Si un type est déclaré dans le package enfant comme 
dérivé de Bill , ce type hérité peut manipuler directement les composants de Bill .

package Orders.From_Home is 
   type Address is new String (1 .. 120); 
 
   type Ordered_By_Phone is new Bill with private; 
 
   procedure Deliver 
     (Ordered : in out Ordered_By_Phone; 
      Place   : in     Address); 
 
private 
   type Ordered_By_Phone is new Bill with 
      record 
         Delivered : Boolean := False; 
         To        : Address; 
      end record; 
end Orders.From_Home;

Orders.From_Home est un package enfant de Orders . Type Ordered_By_Phone est dérivé de Bill et 
inclut son composant d'enregistrement Sum .

Lire Paquets en ligne: https://riptutorial.com/fr/ada/topic/7322/paquets

Chapitre 9: Sortie de nombres

Introduction

Les packages standard d'Ada permettent de générer tous les types numériques. Le format de 
sortie peut être ajusté de plusieurs manières.

Remarques

Notez comment chaque fois qu'un package générique est instancié avec un type numérique. En 
outre, il est possible de définir les deux valeurs par défaut pour l’instance entière et de modifier la 
Width , par exemple, lors de l’appel à Put avec ce paramètre.

Examples

Imprimer des entiers, en utilisant généreusement l'espace

Les instances de Integer_IO ont une variable de paramètres Default_Width dont le nombre de 
caractères que prendra chaque numéro de sortie.

with Ada.Text_IO;   use Ada.Text_IO; 
 
procedure Print_Integer is 
    subtype Count is Integer range -1_000_000 .. 1_000_000; 
 
    package Count_IO is new Integer_IO (Count); 
    X : Count; 
begin 
    Count_IO.Default_Width := 12; 
 
    X := Count'First; 
    while X < Count'Last loop 
        Count_IO.Put (X); 
        Count_IO.Put (X + 1); 
        New_Line; 
 
        X := X + 500_000; 
    end loop; 
end Print_Integer;

Résultat

    -1000000 
     -500000 
           0 
      500000

Impression d'entiers à l'aide de la base 16 (hexadécimal)

Une variable de paramètres Default_Base est définie sur l'instance d' Ada.Text_IO.Integer_IO ; De 
plus, Default_Width est défini pour que la sortie ne puisse pas avoir d'espace de début.

with Ada.Text_IO;   use Ada.Text_IO; 
 
procedure Print_Hex is 
    subtype Count is Integer range -1_000_000 .. 1_000_000; 
 
    package Count_IO is new Integer_IO (Count); 
    X : Count; 
begin 
    Count_IO.Default_Width := 1; 
    Count_IO.Default_Base := 16; 
 
    X := Count'First; 
    while X < Count'Last loop 
        Count_IO.Put (X); 
        New_Line; 
 
        X := X + 500_000; 
    end loop; 
end Print_Hex;

Résultat

-16#F4240# 
-16#7A120# 
16#0# 
16#7A120#

Imprimer des nombres décimaux fixes, alias Money

Ada.Text_IO.Editing offre la mise en forme de valeurs décimales à l'aide de «chaînes d'images». 
Celles-ci décrivent la sortie en utilisant des caractères «magiques» pour les séparateurs, les 
signes monétaires, etc.

with Ada.Text_IO.Editing;   use Ada.Text_IO; 
 
procedure Print_Value is 
 
    Max_Count      : constant := 1_000_000; 
 
    type Fruit is (Banana, Orange, Pear); 
    subtype Count is Integer range -Max_Count .. +Max_Count; 
 
    type Money is delta 0.001 digits 10; 
 
    package Fruit_IO is new Enumeration_IO (Fruit); 
    package Money_IO is new Editing.Decimal_Output 
      (Money, 
       Default_Currency => "CHF", 

       Default_Separator => '''); 
 
    Inventory : constant array (Fruit) of Count := 
      (Banana => +27_420, 
       Orange => +140_600, 
       Pear   => -10_000); 
 
    Price_List : constant array (Fruit) of Money := 
      (Banana => 0.07, 
       Orange => 0.085, 
       Pear   => 0.21); 
 
    Format : constant Editing.Picture := 
      Editing.To_Picture ("<###BZ_ZZZ_ZZ9.99>"); 
begin 
    Fruit_IO.Default_Width := 12; 
 
    for F in Inventory'Range loop 
        Fruit_IO.Put (F); 
        Put          (" | "); 
        Money_IO.Put (Item => Inventory (F) * Price_List (F), 
                      Pic => Format); 
        New_Line; 
    end loop; 
end Print_Value;

Résultat

BANANA       |  CHF     1'919.40 
ORANGE       |  CHF    11'951.00 
PEAR         | (CHF     2'100.00)

Imprimer plusieurs éléments sur une seule ligne

Combinez les instances des packages _IO , utilisez la bonne avec son type numérique.

with Ada.Text_IO;   use Ada.Text_IO; 
 
procedure Print_Inventory is 
    type Fruit is (Banana, Orange, Pear); 
    subtype Count is Integer range -1_000_000 .. 1_000_000; 
 
    package Fruit_IO is new Enumeration_IO (Fruit); 
    package Count_IO is new Integer_IO (Count); 
 
    Inventory : constant array (Fruit) of Count := 
      (Banana => 27_420, 
       Orange => 140_600, 
       Pear   => -10_000); 
 
begin 
    Fruit_IO.Default_Width := 12; 
 
    for F in Inventory'Range loop 
        Fruit_IO.Put (F); 
        Put          (" | "); 

        Count_IO.Put (Inventory (F)); 
        New_Line; 
    end loop; 
end Print_Inventory;

Résultat

BANANA       |    27420 
ORANGE       |   140600 
PEAR         |   -10000

Lire Sortie de nombres en ligne: https://riptutorial.com/fr/ada/topic/8940/sortie-de-nombres

Chapitre 10: Tâche

Syntaxe

tâche nom_tâche;•
task TaskName est la fin des entrées;•
le corps de la tâche Task_Name est les déclarations commencent fin du code;•

Examples

Une tâche simple

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   task My_Task; 
   task body My_Task is 
   begin 
      Put_Line ("Hello from My_Task"); 
   end; 
begin 
   Put_Line ("Hello from Main"); 
end;

Résultat

L'ordre de Put_Line peut varier.

Hello from My_Task 
Hello from Main

Une tâche simple et une boucle

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   task My_Task; 
   task body My_Task is 
   begin 
      for I in 1 .. 4 loop 
         Put_Line ("Hello from My_Task"); 
      end loop; 
   end; 
begin 
   Put_Line ("Hello from Main"); 
end;

Résultat

L'ordre de Put_Line peut varier.

Hello from My_Task 
Hello from Main 
Hello from My_Task 
Hello from My_Task 
Hello from My_Task

Une tâche simple et deux boucles

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   task My_Task; 
   task body My_Task is 
   begin 
      for I in 1 .. 4 loop 
         Put_Line ("Hello from My_Task"); 
      end loop; 
   end; 
begin 
   for I in 1 .. 4 loop 
      Put_Line ("Hello from Main"); 
   end loop; 
end;

Résultat

L'ordre de Put_Line peut varier.

Hello from My_Task 
Hello from My_Task 
Hello from Main 
Hello from My_Task 
Hello from Main 
Hello from My_Task 
Hello from Main 
Hello from Main

Deux tâches simples et deux boucles

with Ada.Text_IO; use Ada.Text_IO; 
 
procedure Main is 
   task My_Task_1; 
   task My_Task_2; 
 
   task body My_Task_1 is 
   begin 

      for I in 1 .. 4 loop 
         Put_Line ("Hello from My_Task_1"); 
      end loop; 
   end; 
 
   task body My_Task_2 is 
   begin 
      for I in 1 .. 4 loop 
         Put_Line ("Hello from My_Task_2"); 
      end loop; 
   end; 
begin 
   null; 
end;

Résultat

L'ordre de Put_Line peut varier.

Hello from My_Task_1 
Hello from My_Task_1 
Hello from My_Task_2 
Hello from My_Task_1 
Hello from My_Task_2 
Hello from My_Task_1 
Hello from My_Task_2 
Hello from My_Task_2

Une tâche qui incrémente un nombre après la saisie

L'utilisateur peut appeler le nombre de fois Incrementor.Increment K en appuyant sur une touche 
dans '0' .. '9' et il est possible d'appeler Incrementor.Increment plus rapidement que la task 
Incrementor peut incrémenter I

with Ada.Text_IO; 
with Ada.Integer_Text_IO; 
 
procedure Main is 
   use Ada.Text_IO; 
   task Incrementor is 
      entry Increment; 
   end; 
   task body Incrementor is 
      use Ada.Integer_Text_IO; 
      I : Integer := 0; 
   begin 
      loop 
         accept Increment; 
         I := I + 1; 
         Put (I, 0); 
         delay 0.1; 
      end loop; 
   end; 
   K : Character; 
begin 

   loop 
      Get_Immediate (K); 
      if K in '0' .. '9' then 
         for I in 1 .. Natural'Value (K & "") loop 
            Incrementor.Increment; 
         end loop; 
      end if; 
   end loop; 
end;

Interruption de la manipulation

Les interruptions sont gérées par une procédure protégée sans paramètres.

------------------------------------------------------------------ 
-- Interrupt Counting Package -- 
------------------------------------------------------------------ 
with Ada.Interrupts.Names; use Ada.Interrupts.Names; 
 
package Ctl_C_Handling is 
 
   protected CTL_C_Handler is 
      procedure Handle_Int with 
        Interrupt_Handler, 
        Attach_Handler => SIGINT; 
      entry Wait_For_Int; 
   private 
      Pending_Int_Count : Natural := 0; 
   end Ctl_C_Handler; 
 
   task CTL_Reporter is 
      entry Stop; 
   end CTL_Reporter; 
 
end Ctl_C_Handling;

Le corps du package montre le fonctionnement de la procédure protégée. Dans ce cas, un 
booléen n'est pas utilisé dans l'objet protégé car les interruptions arrivent plus rapidement qu'elles 
ne sont traitées. La tâche CTL_Reporter gère les interruptions reçues.

with Ada.Text_IO; use Ada.Text_IO; 
with Ctl_C_Handling; use CTL_C_Handling; 
with Ada.Calendar; use Ada.Calendar; 
 
package body Ctl_C_Handling is 
 
   ------------------- 
   -- CTL_C_Handler -- 
   ------------------- 
 
   protected body CTL_C_Handler is 
 
      ---------------- 
      -- Handle_Int -- 
      ---------------- 
 
      procedure Handle_Int is 

      begin 
         Pending_Int_Count := Pending_Int_Count + 1; 
      end Handle_Int; 
 
      ------------------ 
      -- Wait_For_Int -- 
      ------------------ 
 
      entry Wait_For_Int when Pending_Int_Count > 0 is 
      begin 
         Pending_Int_Count := Pending_Int_Count - 1; 
      end Wait_For_Int; 
 
   end CTL_C_Handler; 
 
   ------------------ 
   -- CTL_Reporter -- 
   ------------------ 
 
   task body CTL_Reporter is 
      type Second_Bin is mod 10; 
      type History is array(Second_Bin) of Natural; 
 
      --------------------- 
      -- Display_History -- 
      --------------------- 
 
      procedure Display_History(Item : History) is 
         Sum : Natural := 0; 
      begin 
         for I in Item'Range loop 
            Put_Line("Second: " & Second_Bin'Image(I) & " :" & Natural'Image(Item(I))); 
            Sum := Sum + Item(I); 
         end loop; 
         Put_Line("Total count: " & Natural'Image(Sum)); 
         New_Line(2); 
      end Display_History; 
 
      One_Second_Count : Natural := 0; 
      Next_Slot : Second_Bin := 0; 
      Next_Second : Time := Clock + 1.0; 
      Ten_Second_History : History := (Others => 0); 
 
   begin 
      loop 
         Select 
            Accept Stop; 
            exit; 
         else 
            select 
               CTL_C_Handler.Wait_For_Int; 
               One_Second_Count := One_Second_Count + 1; 
            or 
               delay until Next_Second; 
               Next_Second := Next_Second + 1.0; 
               Ten_Second_History(Next_Slot) := One_Second_Count; 
               Display_History(Ten_Second_History); 
               Next_Slot := Next_Slot + 1; 
               One_Second_Count := 0; 
            end Select; 
         end Select; 

      end loop; 
   end CTL_Reporter; 
end Ctl_C_Handling;

Un exemple de programme principal pour exercer ce paquet est:

------------------------------------------------------------------ 
-- Ada2012 Interrupt Handler Example -- 
------------------------------------------------------------------ 
with Ada.Text_IO; use Ada.Text_IO; 
with Ctl_C_Handling; use CTL_C_Handling; 
 
procedure Interrupt01 is 
begin 
   Delay 40.0; 
   CTL_Reporter.Stop; 
   Put_Line("Program ended."); 
end Interrupt01;

Lire Tâche en ligne: https://riptutorial.com/fr/ada/topic/7345/tache

Chapitre 11: Types paramétrables

Introduction

Tous les types composites autres que les tableaux peuvent avoir des discriminants, qui sont des 
composants avec des propriétés spéciales. Les discriminants peuvent être de type discret ou de 
type accès. Dans ce dernier cas, le type d'accès peut être un type d'accès nommé ou être 
anonyme. Un discriminant d'un type d'accès anonyme est appelé discriminant d'accès par 
analogie avec un paramètre d'accès.

Examples

Types d'enregistrement discriminés

Dans le cas d'un type d'enregistrement discriminé, certains composants sont appelés 
discriminants et les composants restants peuvent en dépendre. Les discriminants peuvent être 
considérés comme paramétrant le type, et la syntaxe révèle cette analogie. Dans cet exemple, 
nous créons un type qui fournit une matrice carrée avec un paramètre positif comme suit:

type Square(X: Positive) is 
    record 
        S: Matrix(1 .. X, 1 .. X); 
    end record;

Ensuite, pour créer un carré de 3 par 3, appelez simplement Carré comme ceci:

Sq: Square(3);

Structures d'enregistrement variées

Un discriminant d'un type d'enregistrement peut influencer la structure des objets. Un choix de 
composants peut exister dans un objet selon qu'un discriminant a eu une valeur particulière lors 
de la création de l'objet. Pour étayer cette variation, la définition d'un type d'enregistrement inclut 
une distinction par cas qui dépend du discriminant:

type Fruit is (Banana, Orange, Pear); 
 
type Basket (Kind : Fruit) is 
   record 
      case Kind is 
         when Banana => 
            Bunch_Size      : Positive; 
            Bunches_Per_Box : Natural; 
         when Pear | Orange => 
            Fruits_Per_Box  : Natural; 
      end case; 
   end record;

Ensuite, créer une boîte pour les bananes,

Box : Basket (Banana);

L'objet Box possède désormais deux composants d'enregistrement en plus de son discriminant, 
Kind , à savoir Bunch_Size et Bunches_Per_Box .

Lire Types paramétrables en ligne: https://riptutorial.com/fr/ada/topic/9311/types-parametrables

Chapitre 12: Types scalaires

Introduction

Dans la hiérarchie des types d'Ada, les types élémentaires ont des ensembles de valeurs 
logiquement indivisibles. Parmi ces types figurent les types d'accès (types de pointeurs) et les 
types scalaires. Les types scalaires peuvent être classés en énumération , caractère et numérique 
. Ces types constituent l'objet de ce sujet. Outre les ensembles de valeurs, les types ont un 
ensemble d'opérations applicables aux scalaires respectifs, tels que le successeur ou "+" .

Syntaxe

type … est …1. 

Paramètres

Ellipse Quelle

… (1) recevoir le nom du type

… (2)
recevoir les caractéristiques du type en utilisant des mots-clés: delta , chiffres , 
plage

Remarques

Toutes les définitions de type scalaire, à l'exception de l'entier et des entiers modulaires, peuvent 
inclure une contrainte d' intervalle .

Une contrainte de plage spécifie une limite inférieure et une limite supérieure de l'ensemble de 
valeurs à inclure dans le type. Pour les types à virgule fixe, spécifier une plage est obligatoire: les 
valeurs de ces types seront comprises comme des multiples d'une petite fraction de deux, par 
exemple, de 1/2 5 . Plus ces fractions deviennent petites, plus la représentation est précise, au 
prix d'une fourchette pouvant être représentée à l'aide des bits disponibles.

D'autres aspects des définitions de type peuvent être donnés, tels qu'une Size souhaitée en bits et 
d'autres éléments de représentation. Ada 2012 ajoute des aspects de la programmation basée sur 
des contrats comme Static_Predicate .

Examples

Énumération

type Fruit is (Banana, Orange, Pear); 

Choice : Fruit := Banana;

Un type de caractère est une énumération qui inclut un littéral de caractère:

type Roman_Numeral is 
    ('I', 'V', 'X', 'L', 'C', 'D', 'M', Unknown);`

Entier Singed

type Grade is range 0 .. 15; 
 
B   : Grade := 11; 
C   : Grade := 8; 
Avg : Grade := (B + C) / 2;  -- Avg = 9

Entier modulaire

Ce sont les types de “bid fiddling”. Ils ont aussi des opérateurs logiques, tels que xor , et ils 
«reviennent» à la limite supérieure, à 0 à nouveau.

type Bits is mod 2**24; 
 
L : Bits := 2#00001000_01010000_11001100# or 7;

Point flottant

Un type à virgule flottante est caractérisé par ses chiffres (décimaux) indiquant la précision 
minimale requise.

type Distance is digits 8; 
 
Earth : Distance := 40_075.017;

Point fixe (ordinaire)

Une définition de type point fixe spécifie un delta et une plage. Ensemble, ils décrivent avec 
précision la manière dont les valeurs réelles doivent être approximées car elles sont représentées 
par des puissances de deux, sans utiliser de matériel à virgule flottante.

Shoe_Ounce : constant := 2.54 / 64.0; 
type Thickness is delta Shoe_Ounce range 0.00 .. 1.00; 
 
Strop : Thickness := 0.1;  -- could actually be 0.09375

Point fixe (décimal)

Les types de virgule décimale sont généralement utilisés en comptabilité. Ils se caractérisent à la 

fois par un delta et un nombre de chiffres décimaux. Leurs opérations arithmétiques reflètent les 
règles de la comptabilité.

type Money is delta 0.001 digits 10; 
 
Oil_Price : Money := 56.402; 
Loss      : Money := 0.002 / 3; -- is 0.000

Lire Types scalaires en ligne: https://riptutorial.com/fr/ada/topic/9297/types-scalaires

